 פונקציות אריתמטיות מובנות
atang2(y, x)
ארקטנגנס של y/x ברדיאנים בטווח +- פיי
cos(x)
קוסינוס x (קלט בראדינים)

exp(x)
 חישוב ex
int(x)
ערך שלם תחתון

log(x)
לוגריתם לפי בסיס e
rand()
מספר אקראי בתחום (0,1)

sin(x)
סינוס x (קלט בראדינים)

sqrt(x)
שורש ריבועי

srand(x)

קביעת ערך התחלתי חדש למחולל מספרים אקראיים (new random seed).

תמיכה מובנית בטיפול במחרוזות
  שרשור  מחרוזות נעשה ע"י כתיבת המחרוזות / ביטויים זה לצד זה ללא מפרידים.

למשל התוכנית:

{print NR ":" $0 }
מדפיס את המחרוזת המתקבלת משרשור NR מותמר למחרוזת עם התו ':' ועם תוכן שורת הקובץ. זה נבדל מהתוכנית

 {print NR ,":", $0 }
בכך שאין את הרווחים המודפסים כמפרידים אוטומטיים במקרה של פסיקים.
האופרטור ~ יכול לקבל כל אופרנד מצד ימין (אם יש צורך הוא מותמר למחרוזת).
לדוגמא,  התוכנית הבאה מדפיסה כל שורה שמכילה  מספרים בלבד
# numbs_filter.awk - print lines containing only numbers
BEGIN{  digits = "^[1-9]+[0-9]*$" }
{ for(i=1; i <= NF; i++)
   if ($i !~ digits)
      next
  print
}
פונקציות מחרוזת מובנות
במובא לעיל r  הוא ביטוי רגולרי (או מחרוזת) ו-s הוא מחרוזת.
בשפת awk  ישנן פונקציות מחרוזות מובנות הבאות:
gsub(r,s)

החלף כל מופע של r ב-s ב-0$

מחזיר שלם – מספר ההחלפות שבוצעו
gsub(r,s,t)

החלף כל מופע של r ב-s ב-t
מחזיר שלם – מספר ההחלפות שבוצעו

index(s,t)

מחזיר שלם – את המיקום הראשון של המחרוזת t  ב-s
length(s)
מחזיר שלם – מספר התווים במחרוזת s
match(s,r)
בודק אם s  מכיל תת מחרוזת שמתאימה ל-r.  מחזיר אינדקס או 0. נותן ערך חדש ל-RSTART  ו-RLENGTH.
split(s,a)
מחלק את המחרוזת s  לתוך מערך המחרוזות a, כאשר FS משמש כמפריד ומחזיר שלם שהוא גודל המערך / מספר השדות שזוהו
split(s,a,fs)

sprintf(fmt, exp-list)
מחזיר כמחרוזת את תוצאת פעולת ה-sprintf.

sub(r, s)
מבצע החלפה של המחרוזת הארוכה ביותר הראשון משמאל שמתאים ל- r ב-0$ ב-s.
החזר את מספר ההחלפות שנעשו.

sub(r, s, t)

מבצע החלפה של המחרוזת הארוכה ביותר הראשון משמאל שמתאים ל- r ב-t ב-s.

החזר את מספר ההחלפות שנעשו.

substr(s,p)
  מחזיר את הסיפא של המחרוזת s החל ממקום p
substr(s, p, n)

 מחזיר את תת המחרוזת של s  החל ממקום p באורך n.

דוגמאות

הפקודה 

gsub("11", "22")

תתמיר את "00011100111" ל-"00022100221"

הפקודה

gsub(/1+/,"22")

תתמיר את "00011100111" ל-"000220022"

הפקודה

 i = index("banana", "an")
תציב ל-i את הערך 2.

הפקודה 

i = length("12345")

תציב לתוך i  את הערך 5.

הפקודה

i = match("00011100111111", /1+/)
תציב לתוך i ו-RSTART  את הערך 4 ול-RLENGTH את הערך 3.

הפקודה
split("AAA BB  CCC", a)
תהפוך את a  למערך שערכיו
a[1] = "AAA"

a[2] = "BB"

a[3] = "CCC"

הפקודה
x = sprintf("%8.3f * %8.3f = %8.3f", 3.1,4.2, 3.1*4.2)
תציב ל-x  את הערך " 3.100 *    4.200 =   13.020".
דומה לפקודה הדומה בשפת C, אך המחרוזת הנוצרת מוחזרת כתוצאת פונקציה ולא דרך פרמטר המציינת יעד.
הפקודה 

sub("11", "22")

תתמיר את "00011100111" ל-"00022100111"

הפקודה

sub(/1+/,"22")

תתמיר את "00011100111" ל-"0002200111"

הפקודה
x = substr("123456", 2)

תציב ל-x את המחרוזת "23456".

הפקודה

x = substr("123456", 2, 3)

תציב ל-x את המחרוזת "234".

מערכים
  שפת awk  תומך במערכים של מספרים או מחרוזות באופי כפול כמו משתנים רגילים.  כמו משתנים מערכים אינם מוכרזים ולמעשה אין שום התחיבות לגבי גודלם. גודל מערך ברגע נתון הוא האינדקס המרבי שניגשו אליו. ערך במערך שלא הוצב בו ערך מאותחל אוטומטית כמו משתנה רגיל (אפס או המחרוזת הריקה).
מערך לפי אינדקס שלם הוא מאינדקס 1 עד  ל-n.  תוכן שדה אינדקס 0 של מערך זהה לאינדקס 1. 

תכונה יוצאת דופן יחסית ב-awk  הוא שהאינדקסים של המערך יכולים להיות מחרוזות.
אפשר לראות פקודות כמו

if($1 == "Asia")

     sums["Asia"] += $1

else

if($1 == "Europe")

  sums["Europe"] += $1

וכו'.
  מערכים מסוג זה נקראים Assocative Arrays.

מאחר וגודל המערך הוא דבר גמיש, יש ב-awk מנגנון של לולאה מיוחדת למערכים

for( משתנה in  מערך)
   פעולה

המשתנה יקבל את כל אחד מאפשרויות האינדקסים במערך והפעולה מבוצעת כמספר האינדקסים האפשריים.  הסדר  של הפעולות תלויה במימוש הפנימי של ה-awk ואינו חייב להיות הסדר הטבעי של האינדקסים.   
  לדוגמא, התוכנית הבאה בונה מערך משורת קלט  בקובץ line  המכיל רק את השורה ""ABCDEFGHJKLMNOPQRSTXYZ.
# arr.awk - test arrays
{ a = $0
  n = length(a)
  for(i=0; i <= n; i++)
   x[i] = substr(a,i,1)
  print a
  for( c in x)
     printf("%c",  x[c])
  printf("\n")
  for(i=0; i <= 26; i++)
     printf("%c",  x[i])
  printf("\n")
}
ההרצה תהיה:

awk -f arr.awk line
ABCDEFGHIJKLMNOPQRSTXYZ
QDRESFGHIJTKXLYMAZNAOBPC
AABCDEFGHIJKLMNOPQRSTXYZ
שימו לב שהתווים לא הודפסו בסדר שהם במערך. שורות הפלט ארוכות באחד משורות הקלט בגלל ש-x[0] הוא כפיל של x[1].  בגלל זה התו 'A' מופיע פעמיים בשורות הפלט.
ניתן לשחרר איבר במערך ע"י פקודת ה-delete למשל
for(i in arr)

  delete arr[i]

ישחרר את כל המערך arr.

מערכים רב מימדיים
ל-awk יש תמיכה במערכים רב מימדיים ממומשים באופן עקיף ע"י מערך חד מימדי עם מפרידים בין כל שורה.  התייחסות לאיבר במערך הרב מימדי הוא 

[ אינדקסים מופרדים ע"י פסיקים ].  לדוגמא,
for(i=1; i <= 20; i++)

  for (j=1; j <= 30; j++)

      arr[i,j] = i*30 + j
פונקציות מתכנת
מתכנת יכול להגדיר פונקציות משלו.  הגדרת הפונקציה יכולה להופיע בכל מקום שאפשר לכתוב ביטוי פעולה.  פונקציה מופעלת רק כאשר קוראים לה.
המבנה של הגדרת פונקציה היא
  function שם (רשימת פרמטרים)

  {

            פקודות

  }
לדוגמא, התוכנית הבאה מדפיסה את הגדול יותר משלושת השדות הראשונים בכל שורה של קובץ הקלט:
# max.awk

{ print max($1, max($2,$3))}
  function max(m,n)
  {
        return m > n ? m : n
  }
שיטת העברת הפרמטרים היא מקבילה לאלו של C כלומר שפרמטרים סקלריים מועברים לפי ערך (by value) ופרמטרים מערכיים מועברים לפי כתובת ( (by reference  כלומר שפונקציה נקראית אינה יכולה לשנות ערכי משתנים המועברים כפרמטרים אבל שדות של מערכים כן.  

   מאחר ואין פוינטרים ב-awk, כאשר פונקציה מחזירה יותר מנתון אחד היא מחזירה אותם דרך מערך, בין אם היה קיים קודם ובין אם לאו.  לדוגמא הפונקציה הבאה min_max מחזירה הן את המקסימום והן את המינימום של מערך נתון:

# minmax1.awk - min and max function.

# Array Parameters are by reference, like C
function min_max(x, result)
{
 result[0] = result[1] = x[0]
 for (i in x)
  {
   if (x[i] > result[0])
     result[0] = x[i]
   else
      if (x[i] < result[1])
            result[1] = x[i]
  } # for
} # min_max
BEGIN {
      x[0] =  23
      x[1] =  12
      x[2] =  31
      x[3] =  56
      x[4] =  11
      x[5] =  43
      y[0] = 0  # Not necessary - for illustration
      y[1] = 0  #    purposes only
      min_max(x, y)
      print "max = ", y[0], "min = ", y[1]
      } # BEGIN
פלט ריצה:

awk -f minmax1.awk

max =  56 min =  11
הפקודות

y[0] = 0 
y[1] = 0

אינן הכרחיות. המערך y כלל לא צריך להיות מוגדר והתוכנית עדיין תעבוד. הן הוכנסו על מנת להמחיש שתוכן מערך המועבר לפרמטר ניתן לשינוי בתוך הפונקציה, כלומר המערך מועבר by reference. 
פלט

הפקודות print ו-printf  יוצרים את הפלט של תוכניות awk.  הפקודה print  היא לפלט פשוט  ו-printf לפלט מורכב ומדויק יותר.  פלט יכול להיכתב לקובץ במקום למסך.
סוגי פקודות פלט:

print
מדפיסה את $0 למסך
print ...,ביטוי, ביטוי
מדפיסה את המרות הביטויים למסך

print ...,ביטוי, ביטוי > שם קובץ 
מדפיסה המרות הביטויים לקובץ, תוך הריסת קובץ קיים.
print ...,ביטוי, ביטוי>>    שם קובץ
מדפיסה את המרות הביטויים לסוף הקובץ במקום להרוס אותו
print ...,ביטוי, ביטוי | פקודה

הפלט מועבר כקלט לפקודה
printf(frmt, ביטוי, ביטוי... )
הדפסה לפי פורמט

printf(frmt, ביטוי, ביטוי... ) >  שם קובץ

printf(frmt, ביטוי, ביטוי... ) >>  שם קובץ
printf(frmt, ביטוי, ביטוי... ) |   פקודה
close(שם קובץ)
סגירת ערוץ לקובץ.  

close(פקודה)

סגירת ערוץ לפקודה.
system(פקודה)
בצע פקודה.  מחזיר את סטאטוס הסיום.
printf  דומה מאד לפקודה המקבילה ב-C למעט שהאופציה * במחרוזת הפורמט אינה נתמכת ושהסוגריים אינם הכרחיים.
 ההפניה לקובץ < פותח את הקובץ רק פעם אחת.  המשך ההדפסות מוזרמות לאותו קובץ. << נבדל ממנה בכך שאם יש קובץ קיים בשם זה התוכן שלו נשמר והפלט החדש רק מתוסף אליו.  למשל אם אנחנו רוצים לפצל קלט מספרי, לקובץ זוגיים ולקובץ אי זוגיים התוכנית הבאה תעשה זאת:
# even_odd.awk

{ if ($1 % 2 == 0)
      print $1 > "even"
  else
      print $1 > "odd"
}
הפנית פלט לפקודה אחרת pipes
  ניתן להפנות פלט של תוכנית awk לפקודות command line  מעשית לתוכנית אחרות. צורת ההפניה
print  ... | "  פקודה"

                                 או
printf ... | "  פקודה"

  לדוגמא,  התוכנית הבאה תדפיס קלט מספרי זוגי באופן ממוין:
# even.awk

{ if ($1 % 2 == 0)
      print $1 | "sort -n"
}
אם התוכנית תורץ על קלט

12449
6767678
667
878
65456
3433
3232
הפלט יהיה:

awk -f even.awk ints

878
3232
65456
6767678
בכדי להבין את המנגנון הזה צריך קצת להבין את הגישה של standard input/standard output ב-C ובמערכות התומכות בגישה הזו.  ב-C (במקור ב-UNIX) היו הרבה תוכניות שמתוכנתות כך שהם מקבלות קלט טקסטואלי או מהמקלדת או מקובץ טקסט או מפלט טקסטואלי של תוכנית אחרת.  הפלט של התוכניות הללו הוא בדרך כלל טקסטואלי והיעד הוא או המסך, או קובץ טקסט או קלט לתוכנית אחרת.  התוכניות אינן מבדילות בן סוגי המקור של הקלט והיעד של הפלט, זה נקבע ע"י הקוד שקורא לתוכנות הללו. כאשר הפלט של תוכנית אחת הוא הקלט של תוכנית אחרת המנגנון שמדובר בו נקרא pipe  והוא מצוין בתוכנית awk כמו גם בשורת הפקודה ע"י התו '|'.
כמו בקבצים, ל-pipe יש פתיחה אוטומטית אחת בלבד.

סגירת ערוצים close
  סגירת ערוצים היא הכרחית אם רוצים לכתוב מידע לקובץ ולקרוא אותו באותה תוכנית. ישנם גם הגבלות מערכת  על מספר הערוצים הפתוחים.
נוסף לכך יש הרבה תוכניות שצוברות את הקלט ועושות את פעולתן רק עם סיומו. 
למשל התוכנית sort.  אם אנחנו רוצים שהמיון בתוכנית even.awk  יבצע את פעולתו בנקודת אמצע של התוכנית יש לבצע משהו כמו 

close("sort –n")
אם אנחנו מוציאם פלט לקובץ בשם "file.txt" כתיבה וסגירה של הקובץ יכול להראות בערך כך:
....

print  > "file.txt"

.....

close("file.txt")

קלט
תוכנית awk  יכולה לקבל קלט בכמה צורות.  
הצורה הנפוצה ביותר היא להעביר שמות קבצים בשורת הפקודה כלומר

awk –f  program  in1.dat in2.dat

כאן in1.dat ו-in2.dat יקראו  לפי הסדר ויועברו כקלט ל-program.

תוכנית awk  יכולה להיות היעד של pipe והפלט של התוכנית הראשונה תהיה הקלט של התוכנית awk למשל
ls  | awk –f program

מפרידי קלט
  המשתנה המובנה FS מכיל את המפריד בין השדות בכל שורה.  ערך ברירת המחדל של המשתנה הזה הוא רווח יחיד.  כאשר זה הערך של FS  ה-awk מתעלם מרווחים נוספים ומטאבים.  שינוי של FS לערך אחר יבטל את המאפיין הנוח בדרך כלל הזה.
שורות או באופן כללי רשומות של מידע מופרדים על ידי תוכן המשתנה המובנה RS שערך ברירת המחדל שלו הוא '\n' (newline ).  לעיתים יש הגיון לשנות את זה (ולדמות רשומות כלליות יותר).  אפשרות אחת למשל היא שורה ריקה "".  אפשר לעשות פקודה נוסך

BEGIN {RS = "" , FS = "\n"}

 ואז אפשר להתייחס לשורות בקובץ כאל שדות מופרדות לרשומות ע"י שורה ריקה.
פונקצית ה-getline
הפונקציה getline מבצעת פעולה של "קרא שורה".  המשמעות של הפקודה תלויה במידה מסוימת בהקשר שהיא נקראת.  הטבלה הבאה מסכמת את האפשרויות:

getline
קורא שורה מהקלט הסטנדרטי (בדרך כלל נעשה אוטומטית) משנה את 0$, NF, NR, FNR
getline var

קורא שורה מהקלט הסטנדרטי לתוך משתנה,  משנה את var, NR, FNR. 
getline < file

קורא שורה נוספת הקובץ.  משנה את 0$, NF. 
getline var < file

קורא שורה נוספת הקובץ.  משנה את   var, 
cmd | getline

קרא שורה נוספת מה-pipe. משנה את 0$.
cmd | getline var

קרא שורה נוספת מה-pipe. משנה את var.

getline  מחזיר 1 אם הקריאה תקינה, 0 אם סוף הקלט ו-1- אם הקובץ לא קיים.
קישור לתוכניות אחרות
ניתן לקרוא לתוכנית חיצונית מתוך תוכנית awk, כמו שאפשר לעשות זאת מתוכנית C. המחרוזת המועברת היא פקודת command line כלשהיא.  תוכנית ה-awk תיעצר עד שהפקודה תתבצע במלואה.
לדוגמא, התוכנית הבאה מקבלת קלט מספרי, שומר את האי זוגיים בקובץ זמני, ממין את תוכנו, ומדפיס את תוכנו, ומוחק אותו.
# close.awk
{ if ($1 % 2 == 0)
      print $1 | "sort -n > ttt"
}
END {
      close("sort -n > ttt")
      while(getline < "ttt" > 0)
             print

      system("rm ttt")
    }
אם הקלט של התוכנית הזו תהיה 

6767678

12449
667
878
65456
3433
3232
הפלט יהיה
878

3232
65456
6767678
פרמטרים לתוכנית ראשית
   ניתן להעביר פרמטרים לתוכנית ראשית לתוכנית awk, הפרמטרים יהיו אלו שאינם אופציות ל-awk.  בדרך כלל הפרמטרים הללו יפורשו כשמות של קובצי קלט אלא אם יהיו מהצורה var=value.  פרמטר var=value יפרש את value כשם של קובץ קלט כשתורו יגיע ובאותו רגע גם יבצע השמה למשתנה var. המשתנית המובנים ARGC, ARGV יהיו להם אותו משמעות כמו המקבילים ב-C: ARGC יכיל את מספר הפרמטרים ו-ARGV מערך של מחרוזות המכיל את ערכם. 
  פרמטרים לתוכנית ראשית נותנים משמעות רצינית לאפשרות של להריץ תוכנית awk ללא קובץ קלט, גם לא מהמקלדת.  במקרה כזה התוכנית תכיל רק בלוק BEGIN.  למשל הפקודה הבאה מחשבת את המחלק המשותף המרבי (greatest common divider) של שני מספרים נתונים לפי האלגוריתם של אוקלידס.

# gcd.awk

BEGIN {
        if (ARGC < 3 )
        {
          print "usage awk -f gcd.awk n1 n2"
          exit
        }
        n1 = ARGV[1]

        n2 = ARGV[2]
        while (n1 != n2)
        {
         if (n1 < n2)
           n2 = n2 - n1
         else
           n1 = n1 - n2
        }
        print "gcd(", ARGV[1], ",", ARGV[2], ") = ", n1
}
הרצת הפקודה

awk -f gcd1.awk 324 63
תיתן פלט

gcd( 324 , 63 ) =  9

כמו ב-C, מספר הפרמטרים ARGV כולל בתור פרמטר ה-0 את שם התוכנית הנקראית, ו-ARGC מביא אותו בחשבון.  התוכנית הבאה מדפיסה רק את הפרמטרים ומסיימת:

התוכנית הבאה רק תדפיס את הפרמטרים לתוכנית:

#echo.awk

BEGIN { for (i = 0; i < ARGC; i++)
          print "ARGV[" i "] = " ARGV[i]
      }
הרצת הפקודה 

awk -f echo.awk A BB CCC
תיתן פלט

ARGV[0] = awk

ARGV[1] = A
ARGV[2] = BB
ARGV[3] = CCC
במידה ויש גוף פקודות פנימי לתוכניות, כל פרמטר שאינו המחרוזת הריקה תפורש כשם של קובץ קלט.  אם בתוך ה-BEGIN יתבצע השמה של המחרוזת הריקה "" לתוך ARGV[i], ה-awk ידלג על הפרמטר.
הפיכת תוכנית awk  לפקודת command line
ב-Unix התהליך יהיה זהה ליצירת כל פקודה מפקודת command line.  כתיבת הפקודה בתוך קובץ טקסט והפיכתו לביצועי בעזרת הפקודה

chmod +x  שם הקובץ

הפרמטרים לתוכנית ראשית אפשר להעביר בעזרת הצירוף *$.

לדוגמא, נכתוב קובץ בשם demo:

 #!/bin/sh

# demo

awk -f echo.awk  $*
נבצע:

bash-2.05b$ chmod +x demo

bash-2.05b$ ./demo A BB CCC
ARGV[0] = awk
ARGV[1] = A
ARGV[2] = BB
ARGV[3] = CCC

bash-2.05b$ 

התוכנית הבאה ממזגת 2 קבצים ממוינים אם שמותיהם מועברים כפרמטרים לתוכנית ראשית:

# merge1.awk

#

#  Usage:  awk -f merge.awk file1 file2

#

# Input files are assumed to be consisted of number only,

#   one number per line, SORTED, and the program only has to 

#    merge them into one sorted file sent to the standard output

#

BEGIN { 

        first_file = ARGV[1]

        ARGV[1] = NULL # Bypass first file when generating standard input

        first_file_flag = (getline first_file_number < first_file)

      } # BEGIN

# main loop

{

  if ( (first_file_flag > 0) && (first_file_number < $1))

    {

     do 

      {

       if (first_file_number < $1) 

             print first_file_number

       else

          break

      } while( (first_file_flag = \

           (getline first_file_number < first_file) ) > 0 )

    } # if

  print $1

} # main loop

END {

     if (first_file_flag > 0) # first file not exhausted

      do {

          print first_file_number

         } while( (getline first_file_number < first_file) > 0 )

    } # END

לדוגמא, אם הקבצים datafiles/sorted2   ו-datafiles/sorted1
מכילים את התכנים

107

107

123

219

256

321

432

543

546

546

1000

1000

1000
ו-

107

119

123

134

218

345

345

546

546
בהתאמה אזי הפקודה:

awk -f merge1.awk datafiles/sorted2 datafiles/sorted1
תמזג אותם לקובץ:

107

107

107

119

123

123

134

218

219

256

321

345

345

432

543

546

546

546

546

1000

1000

1000
